// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/TokenVesting.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.18;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/access/Ownable.sol";

/**
 * @title TokenVesting
 */
contract TokenVesting is Ownable, ReentrancyGuard {
    IERC20 private immutable token;
    uint32 private immutable start;
    uint16 constant SLICE_PERIOD_DAYS = 30; // duration of a slice period for the vesting in days

    struct VestingSchedule {
        uint8 cliffDays;
        uint16 durationDays; // duration of the vesting period in days
        uint112 amountTotal; // total amount of tokens WITHOUT! amountAfterCliff to be released at the end of the vesting
        uint112 released; // amount of tokens released
        //
        uint112 amountAfterCliff;
    }

    mapping(address => VestingSchedule) private vestingSchedules;

    event Clamed(address indexed beneficiary, uint256 amount);
    event ScheduleCreated(
        address indexed beneficiary,
        uint16 durationDays,
        uint112 amount
    );
    event WithdrawedByAdmin(uint256 amount);

    /**
     * @dev Creates a vesting contract.
     * @param _token address of the ERC20 token contract
     * @param _start start timestamp of vesting
     */
    constructor(address _token, uint32 _start) {
        require(_token != address(0x0));
        token = IERC20(_token);
        start = _start;
    }

    /**
     * @notice Creates a new vesting schedule for a beneficiary.
     * @dev _durationDays has 16bits long which could potentially lead to overflow.
     * However, this will only happen if specify a very large number of days, more than 49 711
     * which is more than 139 years. Vesting is designed for a maximum period of 5 years.
     * The threat of overflow is theoretical.
     * @dev It is possible to create flat sloped vesting schedules for long vesting periods with low amounts.
     * There are some border scenarios where the slope calculation outputs zero.
     * But in this vesting it is not planned to use small amounts. All phases are calculated in millions of tokens.
     * But even a single token with 8 decimals will not result in an error.
     * @param _beneficiary address of the beneficiary to whom vested tokens are transferred
     * @param _durationDays duration in days of the period in which the tokens will vest
     * @param _cliffDays duration in days of cliff
     * @param _amountTotal total amount of tokens to be released at the end of the vesting
     * @param _amountAfterCliff amount after cliff
     */
    function createVestingSchedule(
        address _beneficiary,
        uint16 _durationDays,
        uint8 _cliffDays,
        uint112 _amountTotal,
        uint112 _amountAfterCliff
    ) external onlyOwner {
        require(
            start > uint32(block.timestamp),
            "TokenVesting: forbidden to create a schedule after the start of vesting"
        );
        require(_durationDays > 0, "TokenVesting: duration must be > 0");
        require(_amountTotal > 0, "TokenVesting: amount must be > 0");
        require(
            _durationDays >= uint16(_cliffDays),
            "TokenVesting: duration must be >= cliff"
        );
        vestingSchedules[_beneficiary] = VestingSchedule(
            _cliffDays,
            _durationDays,
            _amountTotal,
            0,
            _amountAfterCliff
        );

        token.transferFrom(
            msg.sender,
            address(this),
            _amountTotal + _amountAfterCliff
        );

        emit ScheduleCreated(
            _beneficiary,
            _durationDays,
            _amountTotal + _amountAfterCliff
        );
    }

    /**
     * @notice claim vested amount of tokens.
     */
    function claim() external nonReentrant {
        VestingSchedule storage vestingSchedule = vestingSchedules[msg.sender];
        require(
            vestingSchedule.amountTotal > 0,
            "TokenVesting: only investors can claim"
        );
        uint256 vestedAmount = _computeReleasableAmount(vestingSchedule);
        require(vestedAmount > 0, "TokenVesting: nothing to claim");

        // amountAfterCliff - could not be > vestedAmount, because it used in calculation of vestedAmount
        vestingSchedule.released += (uint112(vestedAmount) -
            vestingSchedule.amountAfterCliff);
        vestingSchedule.amountAfterCliff = 0;
        token.transfer(msg.sender, vestedAmount);

        emit Clamed(msg.sender, vestedAmount);
    }

    /**
     * @notice Computes the vested amount of tokens for the given vesting schedule identifier.
     * @return the vested amount
     */
    function computeReleasableAmount(
        address _beneficiary
    ) external view returns (uint256) {
        return _computeReleasableAmount(vestingSchedules[_beneficiary]);
    }

    /**
     * @notice Returns the vesting schedule information for a given identifier.
     * @return the vesting schedule structure information
     */
    function getVestingSchedule(
        address _beneficiary
    ) external view returns (VestingSchedule memory) {
        return vestingSchedules[_beneficiary];
    }

    /**
     * @dev Computes the releasable amount of tokens for a vesting schedule.
     * @return the amount of releasable tokens
     */
    function _computeReleasableAmount(
        VestingSchedule memory vestingSchedule
    ) internal view returns (uint256) {
        // If the current time is before the cliff, no tokens are releasable.
        uint32 cliffDuration = (uint32(vestingSchedule.cliffDays) * 1 days);
        if (uint32(block.timestamp) < start + cliffDuration) {
            return 0;
        }
        // If the current time is after the vesting period, all tokens are releasable,
        // minus the amount already released.
        else if (
            uint32(block.timestamp) >=
            start + (uint32(vestingSchedule.durationDays) * 1 days)
        ) {
            return
                uint256(
                    vestingSchedule.amountTotal +
                        vestingSchedule.amountAfterCliff -
                        vestingSchedule.released
                );
        }
        // Otherwise, some tokens are releasable.
        else {
            uint32 vestedSlicePeriods = (uint32(block.timestamp) -
                start -
                cliffDuration) / (uint32(SLICE_PERIOD_DAYS) * 1 days); // Compute the number of full vesting periods that have elapsed.
            uint32 vestedSeconds = vestedSlicePeriods *
                (uint32(SLICE_PERIOD_DAYS) * 1 days);
            uint256 vestedAmount = (vestingSchedule.amountTotal *
                uint256(vestedSeconds)) /
                ((uint256(vestingSchedule.durationDays) -
                    uint256(vestingSchedule.cliffDays)) * 1 days); // Compute the amount of tokens that are vested.
            return
                vestedAmount +
                uint256(vestingSchedule.amountAfterCliff) -
                uint256(vestingSchedule.released); // Subtract the amount already released and return.
        }
    }
}